

### Multivariate Models for Dyadic and Family Data

These data were sampled from a study of family dynamics conducted at Penn State University. We will examine the marital gender attitudes for an adult child and his/her mother and father as a function of education.

```

** Bringing in stacked data from SPSS for mom, dad, kid;
DATA family; SET import.&datafile.;
    * Indicators for persons;
    IF target=2 THEN DO; kid=1; mom=0; dad=0; parent=0; END;
    ELSE IF target=3 THEN DO; kid=0; mom=1; dad=0; parent=1; END;
    ELSE IF target=4 THEN DO; kid=0; mom=0; dad=1; parent=1; END;
    LABEL kid=      "Offspring (0=no, 1=yes)"
           mom=     "Mother (0=no, 1=yes)"
           dad=     "Father (0=no, 1=yes)"
           DadEd12="Dad's Years of Education (0=12)"
           MomEd12="Mom's Years of Education (0=12)"
           KidEd12="Kid's Years of Education (0=12)"
           Marit=   "Marital Gender Attitudes Mean (1-4)"; RUN;
    
```

In this example, the unit of analysis is family, and the multivariate outcomes will be responses from different persons from the same family, which will be noted as *kid*, *mom*, and *dad* below. Rather than directly estimating differences between DVs (people here) directly as we've done before, this example will demonstrate how to fit multivariate models in which the effects are estimated *separately* per DV. This approach will have some unique advantages in modeling dyadic and family data in which not all effects may need to be included for each person.

#### Empty Multivariate Model Predicting Conservative Marital Attitudes (using Multivariate Normal Distribution and Identity Link): This model predicts attitudes for person *p* in family *f* →

$$Att_{fp} = Kid[\beta_{0k} + e_{fk}] + Mom[\beta_{0m} + e_{fm}] + Dad[\beta_{0d} + e_{fd}]$$

```

TITLE "Empty Multivariate Model for Marital Attitudes";
PROC MIXED DATA=family COVTEST IC NAMELEN=50 METHOD=ML;
    CLASS target;
    MODEL marit = kid mom dad / NOINT SOLUTION DDFM=KR;
    REPEATED target / R RCORR TYPE=UN SUBJECT=familyID;
    ESTIMATE "Kid vs. Mom: Intercept Diff"      kid -1 mom 1 dad 0;
    ESTIMATE "Kid vs. Dad: Intercept Diff"      kid -1 mom 0 dad 1;
    ESTIMATE "Mom vs. Dad: Intercept Diff"      kid 0 mom -1 dad 1;
    ESTIMATE "Kid vs. Parents: Intercept Diff"  kid -1 mom .5 dad .5;
RUN;
    
```

Because the DVs are persons with no logical order, I'm going with unstructured as a default.

Estimated R Matrix for Subject 1				Estimated R Correlation Matrix for Subject 1			
Row	Col1	Col2	Col3	Row	Col1	Col2	Col3
1	0.3356	0.04019	0.08500	1	1.0000	0.1239	0.2593
2	0.04019	0.3135	0.09175	2	0.1239	1.0000	0.2896
3	0.08500	0.09175	0.3202	3	0.2593	0.2896	1.0000

#### Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z	
UN(1,1)	FamilyID	0.3356	0.03941	8.51	<.0001	Variance across Kids
UN(2,1)	FamilyID	0.04019	0.02718	1.48	0.1392	Kid-Mom Covariance
UN(2,2)	FamilyID	0.3135	0.03697	8.48	<.0001	Variance across Moms
UN(3,1)	FamilyID	0.08500	0.02873	2.96	0.0031	Kid-Dad Covariance
UN(3,2)	FamilyID	0.09175	0.02763	3.32	0.0009	Mom-Dad Covariance
UN(3,3)	FamilyID	0.3202	0.03810	8.40	<.0001	Variance across Dad

#### Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
711.8	9	729.8	730.3	740.7	756.6	765.6

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
kid	1.6411	0.04811	145	34.11	<.0001
mom	1.9043	0.04664	144	40.83	<.0001
dad	1.9649	0.04757	142	41.31	<.0001

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Kid vs. Mom: Intercept Diff	0.2632	0.06274	145	4.19	<.0001
Kid vs. Dad: Intercept Diff	0.3237	0.05836	142	5.55	<.0001
Mom vs. Dad: Intercept Diff	0.06057	0.05635	144	1.07	0.2842
Kid vs. Parents: Intercept Diff	0.2935	0.05364	144	5.47	<.0001

**Adding effect of gender only for adult offspring (kid) because is redundant otherwise;  
Adding effects of own education on own attitudes**

$$Att_{fp} = Kid[\beta_{0k} + \beta_{1k}Gender_{fp} + \beta_{2k}KidEduc_{fk} + e_{fk}] + Mom[\beta_{0m} + \beta_{2m}MomEduc_{fm} + e_{fm}] + Dad[\beta_{0d} + \beta_{2d}DadEduc_{fd} + e_{fd}]$$

```
TITLE "Adding Kid Gender and Own Education";
PROC MIXED DATA=family COVTEST IC NAMELEN=50 METHOD=ML;
CLASS target;
MODEL marit = kid mom dad kid*KidBoy kid*KidEd12 mom*MomEd12 dad*DadEd12
/ NOINT SOLUTION DDFM=KR;
REPEATED target / R RCORR TYPE=UN SUBJECT=familyID;
ESTIMATE "Kid vs. Mom: Diff in Own Ed Effect" kid*KidEd12 -1 mom*MomEd12 1 dad*DadEd12 0;
ESTIMATE "Kid vs. Dad: Diff in Own Ed Effect" kid*KidEd12 -1 mom*MomEd12 0 dad*DadEd12 1;
ESTIMATE "Mom vs. Dad: Diff in Own Ed Effect" kid*KidEd12 0 mom*MomEd12 -1 dad*DadEd12 1;
ESTIMATE "Kid vs. Parent: Diff in Own Ed Effect" kid*KidEd12 -1 mom*MomEd12 .5 dad*DadEd12 .5;
RUN;
```

Neg2LogLike	Parms	Information Criteria				
		AIC	AICC	HQIC	BIC	CAIC
691.1	13	717.1	717.9	732.8	755.8	768.8

Is this a better model?

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
kid	1.5326	0.09829	148	15.59	<.0001
mom	1.9387	0.05817	148	33.33	<.0001
dad	2.0805	0.05641	149	36.88	<.0001
kid*KidBoy	0.2713	0.09158	145	2.96	0.0036
kid*KidEd12	-0.00673	0.02344	146	-0.29	0.7745
mom*MomEd12	-0.01660	0.01689	148	-0.98	0.3272
dad*DadEd12	-0.05530	0.01581	146	-3.50	0.0006

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Kid vs. Mom: Diff in Own Ed Effect	-0.00987	0.02850	233	-0.35	0.7293
Kid vs. Dad: Diff in Own Ed Effect	-0.04857	0.02728	209	-1.78	0.0765
Mom vs. Dad: Diff in Own Ed Effect	-0.03870	0.02137	186	-1.81	0.0717
Kid vs. Parent: Diff in Own Ed Effect	-0.02922	0.02577	195	-1.13	0.2582

**Adding effects of Dad's education on other attitudes (controlling for own education)**

$$\begin{aligned}
 Att_{fp} = & Kid[\beta_{0k} + \beta_{1k}Gender_{fp} + \beta_{2k}KidEduc_{fk} + \beta_{3k}DadEduc_{fk} + e_{fk}] \\
 & + Mom[\beta_{0m} + \beta_{2m}MomEduc_{fm} + \beta_{3m}DadEduc_{fm} + e_{fm}] \\
 & + Dad[\beta_{0d} + \beta_{2d}DadEduc_{fd} + e_{fd}]
 \end{aligned}$$

```

TITLE "Adding Effects of Dads' Education";
PROC MIXED DATA=family COVTEST IC NAMELEN=50 METHOD=ML;
CLASS target;
MODEL marit = kid mom dad kid*KidBoy kid*KidEd12 mom*MomEd12 dad*DadEd12
            kid*DadEd12 mom*DadEd12 / NOINT SOLUTION DDFM=KR;
REPEATED target / R RCORR TYPE=UN SUBJECT=familyID;
ESTIMATE "Kid vs. Mom: Diff in Dad Ed Effect" kid*DadEd12 -1 mom*DadEd12 1 dad*DadEd12 0;
ESTIMATE "Kid vs. Dad: Diff in Dad Ed Effect" kid*DadEd12 -1 mom*DadEd12 0 dad*DadEd12 1;
ESTIMATE "Mom vs. Dad: Diff in Dad Ed Effect" kid*DadEd12 0 mom*DadEd12 -1 dad*DadEd12 1;
ESTIMATE "Kid vs. Parent: Diff in Dad Ed Effect" kid*DadEd12 -1 mom*DadEd12 .5 dad*DadEd12 .5;
RUN;
    
```

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
690.8	15	720.8	721.9	738.9	765.4	780.4

Solution for Fixed Effects

Is this a better model?

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
kid	1.5416	0.09968	149	15.47	<.0001
mom	1.9404	0.06068	145	31.98	<.0001
dad	2.0846	0.05717	142	36.46	<.0001
kid*KidBoy	0.2676	0.09181	145	2.91	0.0041
kid*KidEd12	-0.00298	0.02456	144	-0.12	0.9036
mom*MomEd12	-0.01673	0.02045	144	-0.82	0.4146
dad*DadEd12	-0.05728	0.01644	143	-3.48	0.0007
kid*DadEd12	-0.00904	0.01744	147	-0.52	0.6051
mom*DadEd12	-0.00069	0.01990	148	-0.03	0.9722

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Kid vs. Mom: Diff in Dad Ed Effect	0.008343	0.02521	172	0.33	0.7411
Kid vs. Dad: Diff in Dad Ed Effect	-0.04825	0.02093	150	-2.30	0.0226
Mom vs. Dad: Diff in Dad Ed Effect	-0.05659	0.02287	164	-2.47	0.0144
Kid vs. Parent: Diff in Dad Ed Effect	-0.01995	0.02015	164	-0.99	0.3236

So much for the theory that dad behavior shapes attitudes! What about moms?

**Instead adding effects of Mom's education on other attitudes (controlling for own education)**

$$Att_{fp} = Kid[\beta_{0k} + \beta_{1k}Gender_{fp} + \beta_{2k}KidEduc_{fk} + \beta_{4k}MomEduc_{fk} + e_{fk}]$$

$$+ Mom[\beta_{0m} + \beta_{2m}MomEduc_{fm} + e_{fm}]$$

$$+ Dad[\beta_{0d} + \beta_{2d}DadEduc_{fd} + \beta_{3d}MomEduc_{fd} + e_{fd}]$$

```
TITLE "Adding Effects of Moms' Education";
PROC MIXED DATA=family COVTEST IC NAMELEN=50 METHOD=ML;
  CLASS target;
  MODEL marit = kid mom dad kid*KidBoy kid*KidEd12 mom*MomEd12 dad*DadEd12
            kid*MomEd12 dad*MomEd12 / NOINT SOLUTION DDFM=KR;
  REPEATED target / R RCORR TYPE=UN SUBJECT=familyID;
ESTIMATE "Kid vs. Mom: Diff in Mom Ed Effect" kid*MomEd12 -1 mom*MomEd12 1 dad*MomEd12 0;
ESTIMATE "Kid vs. Dad: Diff in Mom Ed Effect" kid*MomEd12 -1 mom*MomEd12 0 dad*MomEd12 1;
ESTIMATE "Mom vs. Dad: Diff in Dad Ed Effect" kid*MomEd12 0 mom*MomEd12 -1 dad*MomEd12 1;
ESTIMATE "Kid vs. Parent: Diff in Mom Ed Effect" kid*MomEd12 -1 mom*MomEd12 .5 dad*MomEd12 .5;
RUN;
```

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
681.8	15	711.8	713.0	730.0	756.5	771.5

Is this a better model than just predicting with one's own education?

Is this a better model than predicting with dad's education instead?

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
kid	1.5407	0.09885	150	15.59	<.0001
mom	1.9590	0.05848	144	33.50	<.0001
dad	2.1347	0.05854	144	36.46	<.0001
kid*KidBoy	0.2671	0.09099	145	2.94	0.0039
kid*KidEd12	-0.01446	0.02470	145	-0.59	0.5593
mom*MomEd12	-0.02631	0.01716	144	-1.53	0.1274
dad*DadEd12	-0.02445	0.01847	141	-1.32	0.1876
kid*MomEd12	0.008423	0.01834	147	0.46	0.6468
dad*MomEd12	-0.05657	0.01973	149	-2.87	0.0047

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Kid vs. Mom: Diff in Mom Ed Effect	-0.03474	0.02366	153	-1.47	0.1442
Kid vs. Dad: Diff in Mom Ed Effect	-0.06500	0.02374	167	-2.74	0.0069
Mom vs. Dad: Diff in Dad Ed Effect	-0.03026	0.02310	166	-1.31	0.1919
Kid vs. Parent: Diff in Mom Ed Effect	-0.04987	0.02070	164	-2.41	0.0171